

Themenübersicht Dezember 2002

Ausgabe: 12 / 2002

- Nice to know
- Fusspunkterregung bei einer transienten Analyse
- Beulanalyse bei zyklisch symmetrischen Modellen
- Berechnung von Strahlungsproblemen mit dem Radiositysolver
- Berechnung von Ausfallwahrscheinlichkeiten
- Termine rund um CADFEM

- **Unter anderem in der nächsten Ausgabe:**

Substrukturanalyse mit dem Sparse Solver
Neue Anwendungsmöglichkeiten der HF-Elemente

In eigener Sache:

Die Zusendung dieser Informationen erfolgt ausschließlich auf Wunsch des Empfängers und kann jederzeit unter www.cadfem.de beendet werden.

Wenngleich die vorliegenden Informationen mit größter Sorgfalt erstellt worden sind, weisen wir darauf hin, dass die Verwendung dieser unter Ausschluss jeglicher Gewährleistung erfolgt.

Nice to know

Ausgabe: 12 / 2002

ANSYS

- Man kann den Elementtyp auch bei bereits vernetzter Geometrie nachträglich ändern, indem man einfach den entsprechenden Elementtyp überschreibt (ET,1,185 -> ET,1,5).
Die funktioniert problemlos zwischen Elementen mit gleicher Knotenanzahl. Dort werden auch die Randbedingungen entsprechend der neuen Freiheitsgrade richtig übernommen.
Wenn man zwischen Elementen mit unterschiedlicher Knotenanzahl umschalten möchte, muss man zunächst die FE-Struktur von der Geometrie lösen (MODMSH,DETACH) und anschliessend können Mittelknoten eingefügt oder entfernt werden (EMID,ADD bzw. EMID,REMOVE).
- Bei plastischen Rechnungen können sich möglicherweise beim `plns,s,eqv` und `ples,s,eqv` Kommando Spannungen ergeben, die oberhalb der Fließgrenze liegen, obwohl noch keine Plastifizierung eingetreten ist.
Ursache ist die Extrapolation der Spannungen von den Integrationspunkten auf die Knoten, dadurch können höhere Werte als die Fließgrenze angezeigt werden.
Das Kommando `ERESX,NO`
wenn es vor der Lösung ausgeführt wird, sorgt dafür, dass die Ergebnisse nicht extrapoliert werden sondern lediglich von den Integrationspunkten auf die Knoten kopiert werden.
- In der Version 7.0 erscheint auf das `SET,LIST` Kommando nur noch eine Auflistung der Results, die nicht mehr interaktiv bedienbar ist. Die interaktive Resultübersicht erhält man nun mit `SET,LIST,999`.
- Bei einer transienten Rechnung kann mit
`antyp,stat`
`timin,off`
`time,1e-5`
`solve`
`timin,on`
`time,1`
`solve`
zur Stabilisierung eine statische Lösung vorweggeschaltet werden.

DesignSpace

- Mit dem AGP (Analysis Geometry Processor) ist es möglich kleine Modelle vollparametrisch aufzubauen und an ANSYS, Workbench Environment und DesignSpace zu übergeben.
Über die Schnittstellen zu allen gängigen CAD Programmen können CAD Modelle eingeladen und für die FE aufbereitet werden.
Dazu zählt zum Beispiel die Teilung von Flächen um gezielt Randbedingungen aufzubringen.
Ebenso gehört dazu kleine Feature, die beim automatischen Vernetzen unnötig viele Elemente produzieren, zu entfernen.
Es ist auch möglich Körper zu teilen um ein regelmässiges Netz zu erhalten.

Fusspunkterregung bei einer transienten Analyse

Ausgabe: 12 / 2002

Problem:

Willt man auf eine Struktur im Rahmen einer transienten Analyse eine Beschleunigung aufbringen, bieten sich dazu zunächst verschiedene Möglichkeiten an. Es gibt die Möglichkeit einen Beschleunigungsverlauf zu einem Verlauf der Verschiebungen zu integrieren. Als weiteres gibt es die Möglichkeit eine Massenelement an den Körper zu setzen und diese mit einem Kraftverlauf zu belegen. Es stellt sich nun die Frage welche Modellierung unter welchen Voraussetzungen verwendet werden kann und was bei der Modellierung zu berücksichtigen ist.

Erläuterung:

Liegt für eine transiente Analyse die Last in Form der Beschleunigung über der Zeit vor, so stellt sich die Frage wie diese auf einzelne Knoten der FE-Models aufgebracht werden kann. Primär können auf Knoten zunächst nur Kräfte/Momente oder Verschiebungen/Rotationen aufgebracht werden.

Liegt der Beschleunigungs-Zeit-Verlauf vor, so kann man zunächst einmal durch zweifache Integration den Verschiebungs-Zeit-Verlauf ermitteln. Dabei sind als Randbedingungen die Geschwindigkeit und Verschiebung zur Zeit $t=0$ zu berücksichtigen. Die so berechneten Verschiebungen können dann als Belastung auf Knoten über der Zeit aufgebracht werden.

Das funktioniert allerdings nicht, wenn die transiente Analyse mittels der Methode der modalen Superposition durchgeführt wird. Für diesen Fall ist es dann aber möglich, die anzuregenden Knoten mit Massenelementen zu belegen. Wegen $F = ma$ oder auch $a = F/m$ kann dann durch Aufbringen geeigneter Kräfte auf die Knoten eine Beschleunigung modelliert werden. Dabei kann die Größe der Punktmassen angepasst werden, indem man zum Beispiel die Eigenfrequenzen der Modalanalyse bewertet.


Beispiel:

Betrachtet wird ein eingespannter Balken, der am Fußpunkt in X-Richtung beschleunigt werden soll. Die Beschleunigung-Zeit-Funktion liegt vor. Die Maximalbeschleunigung soll $a = 0.2g$ betragen. Mit $g = 9810 \text{ mm/s}^2$ und einer Punktmasse von $m = 7.8e-03 \text{ t}$ ergibt sich für die Kraftamplitude $F = 15303.6e-03 \text{ N}$. Der Balken ist zur Zeit $t = 0$ in Ruhe. Es soll mit der Methode der modalen Superposition gearbeitet werden.

Durch Variieren der Punktmasse findet man heraus, dass sich bei der oben angegebenen Größenordnung die Eigenfrequenzen nicht mehr merklich ändern.

Die Last-Zeit-Funktion für die transiente Analyse wird mit Hilfe einer Tabelle erstellt. Aus diese Tabelle werden dann im Lösungsabschnitt der Transienten die entsprechenden Kraftgrößen mittels einer DO-Schleife herausgelesen.

Im /POST26 können sofort die Verschiebungen einzelner Punkte über der Zeit dargestellt werden. Um einen Konturplot oder das verformte Netz zu diskreten Zeitpunkten darstellen zu können, ist eine Expansion der Ergebnisse nötig.



Fusspunkterregung bei einer transienten Analyse

Ausgabe: 12 / 2002

ANSYS Eingabesatz (ANSYS 7.0):

```
fini
/clear

/prep7
et,1,beam3
et,2,mass21
keyopt,2,3,4
mp,ex,1,210000
mp,prxy,1,0.3
mp,dens,1,7.8e-9
r,1,3,2.25,3
r,2,7.8e-3
k,1,0,0,0
k,2,0,100,0
l,1,2

esize,10
type,1
real,1
lmesh,all
type,2
real,2
kmesh,1

nset,s,loc,y,0
cm,ziel,node
d,all,uy,0
d,all,rotz,0
allsel

/solu
antype,modal
modopt,lanb,10
mxpa,10
solve

fini
/solu
antype,trans
trnopt,msup,10
outres,all,all
lumpm,on
```

```
*dim,last,table,6,1
!Zahlenwerte der Zeit
last(1,0)=0.0
last(2,0)=1.0e-5
last(3,0)=5.0e-4
last(4,0)=1.5e-3
last(5,0)=2.0e-3
last(6,0)=5.0e-3
!Zahlenwerte der Kraft
last(1,1)=0.0
last(2,1)=0.0
last(3,1)=+15303.6e-3
last(4,1)=-15303.6e-3
last(5,1)=0.0
last(6,1)=0.0

/axlab,x,Zeit
/axlab,y,Kraft
*vplot,last(1,0),last(1,1)

delttime,1.0e-5
f,ziel,fx,0
solve
*do,zeit,1.0e-4,5.0e-3,1.0e-4
time,zeit
f,ziel,fx,last(zeit)
solve
*enddo
time,5.0e-3
f,ziel,fx,last(5.0e-3)
solve

/post26
file,,rdsp
nsol,2,1,u,x,ux_unt
nsol,3,2,u,x,ux_obe
/axlab,x,Zeit
/axlab,y,Weg
plvar,2,3

/solu
expass,on
numexp,20,0.0,5.0e-3
solve
/post1
set,first
/eshape,1
plnsol,u,sum,2,1
antime,20,0.5,,1,2,2.5e-4,5.0e-3
```

Beulanalyse bei zyklisch symmetrischen Modellen

Ausgabe: 12 / 2002

Problem:

In der Version 7.0 ist es möglich mit ANSYS eine Eigenwert Beulanalyse zyklisch symmetrischer Bauteile durchzuführen. Es soll das grundsätzliche Vorgehen dargelegt werden und auf die Grenzen der Methode hingewiesen werden.

Erläuterung:

Für zyklisch symmetrische Modelle bietet ANSYS die Möglichkeit nur einen Sektor des Gesamt-modells zu berechnen und an den Begrenzungsflächen zyklisch symmetrische Randbedingungen aufzubringen. Seit der Version 7.0 kann damit auch das Eigenwertbeulen berechnet werden.

Nachdem bereits vor dem Vernetzen mit dem Befehl CYCLIC ANSYS angewiesen wurde, dass es sich um eine zyklisch symmetrische Analyse handelt und ANSYS daraufhin beim Vernetzen auf gleiche Knotenmuster an den beiden begrenzenden Seitenflächen achtet, entspricht die Analyse dem Vorgehen bei einer normalen Beuluntersuchung. Beim Postprocessing werden die Ergebnisse mit /CYEXPAND auf die vollen 360° expandiert.

Bei einer einfachen Scheibe kann man sich vorstellen, dass die Ergebnisse der Berechnung einer Modalanalyse (diese ist genauso eine Eigenwertanalyse wie das lineare Beulen) auf der Aussenkontour der Scheibe betrachtet eine Schwingung um die Geometrie herum darstellen. Hier wird der Begriff der Knotendurchmesser eingeführt, der die Anzahl der Nulldurchgänge dieser Schwingung repräsentiert. Die Berechnungsmethode legt die Ergebnisse zu jeder Anzahl an Knotendurchmessern in einem Lastschritt ab. Die Anzahl der Ergebnisse im Resultfile ergibt sich damit zu: Anzahl der Knotendurchmesser x Anzahl der angeforderten Moden. Diese Ergebnisse liegen nicht in nach Grösse des Lastniveaus geordneter Form vor. Dies kann mit der Option ORDER im SET Befehl erreicht werden.




Beispiel:

Es wird eine rippenversteifte Scheibe unter einer Einheits-Drucklast betrachtet. Es wird dazu die zyklische Symmetrie ausgenutzt und ein 30° Modell verwendet. Die Ergebnisse aus der linearen Beulanalyse kann man mit denen aus einer Berechnung am Vollmodell vergleichen.

Es ist zwar möglich bei zyklisch symmetrischen Randbedingungen eine geometrisch nichtlineare Rechnung anzustossen, die Aufbringung von Imperfektionen ist allerdings nur unter bestimmten Umständen an einem zyklisch symmetrischen Modell möglich. Will man also eine nichtlineare Beulanalyse durchführen empfiehlt es sich nachwievor auf die Modellierung des Gesamtmodells zurückzugreifen.

Ein weiterer Aspekt: Man kann man nur im Rahmen einer nichtlinearen Stabilitätsuntersuchung den Spannungszustand aus Rotation berücksichtigen, indem man die Rotation in einen vorgeschalteten Lastschritt stellt. Bei der linearen Analyse wird stattdessen ein Lastfaktor berechnet, der die gesamte Belastung (inkl. der Rotationsbelastung) skaliert.

In dem vorliegenden Beispiel erkennt man auch aus der nichtlinearen Beulanalyse, dass das Modell ein gutmütiges Verhalten aufweist. Bei Erreichen der Verzweigungslast stellt sich nicht Stabilitätsversagen ein, sondern es erfolgt eine starke Zunahme der Verformungen bei einer gewissen Reststeifigkeit.



SET	TIME/FREQ	LOAD STEP	SUBSTEP	CUMULATIVE	MOD-INDEX
19	1.6630	7	1	7	6
16	1.6628	6	1	6	5
17	1.6628	6	2	6	5
13	1.6657	5	1	5	4
14	1.6657	5	2	5	4
10	1.7202	4	1	4	3
11	1.7202	4	2	4	3
7	1.7291	3	1	3	2
8	1.7291	3	2	3	2
4	1.7403	2	1	2	1
5	1.7403	2	2	2	1
1	1.7417	1	1	1	0
20	2.2537	7	1	7	6
18	2.2545	6	1	6	5
15	2.2617	5	1	5	4
12	2.2675	4	1	4	3
9	2.3080	3	1	3	2
6	2.3203	2	1	2	1
2	2.3237	1	1	1	0
21	4.1377	7	1	7	6
3	4.2581	1	1	1	0

SET	TIME/FREQ	LOAD STEP	SUBSTEP	CUMULATIVE
1	1.6630	1	1	1
2	1.6628	1	2	1
3	1.6628	1	3	1

Beulanalyse bei zyklisch symmetrischen Modellen

Ausgabe: 12 / 2002

ANSYS Eingabesatz (ANSYS 7.0)

für die lineare Beulanalyse am Sektorenmodell:

```
fini
/clear
/prep7
et,1,181
r,1,.05
```

```
mp,ex,1,210000
mp,prxy,1,0.3
mp,dens,1,7.8e-8
```

```
csys,1
k,1,1,0,0
k,2,1,0,1
k,3,1,15,1
k,4,1,30,1
k,5,1,30,0
k,6,1,15,0
k,7,5,0,0
k,9,5,15,1
k,11,5,30,0
k,12,5,15,0
a,1,2,3,6
a,3,4,5,6
a,3,9,12,6
a,1,7,12,6
a,5,6,12,11
aglu,all
```

```
cyclic
mshkey,1
esize,.3
amesh,all
```

```
/solu
antype,static
asel,s,loc,z,0
alls,below,area
nsla,s,1
sf,all,pres,1
asel,s,loc,x,1
alls,below,area
nsla,s,1
```

```
d,all,ux,0,,,,,uy,zu
pstres,on
alls
cycopt,status
solve
fini
```

```
/solu
antype,buckle
bucopt,lanb,3
pstres,on
outres,all,all
cycopt,status
```

```
alls
solve
/post1
set,list,,,,,,order
set,first,,,,,order
/cycexpand,,on
plns,u,sum
```

Auf der nächsten Seite finden Sie den Inputfile zur nichtlinearen Stabilitätsuntersuchung am Vollmodell.

Beulanalyse bei zyklisch symmetrischen Modellen

Ausgabe: 12 / 2002

ANSYS Eingabesatz (ANSYS 7.0)

für die nichtlineare Beulanalyse am Vollmodell:

```
Finis
/clear
/filna,luefter

/prep7

et,1,181
r,1,.05

mp,ex,1,210000
mp,prxy,1,0.3
mp,dens,1,7.8e-8

csys,1

k,1,1,0,0
k,2,1,0,1
k,3,1,15,1
k,4,1,30,1
k,5,1,30,0
k,6,1,15,0

k,7,5,0,0
k,9,5,15,1
k,11,5,30,0
k,12,5,15,0

a,1,2,3,6
a,3,4,5,6
a,3,9,12,6
a,1,7,12,6
a,5,6,12,11
aglu,all

mshkey,1
esize,.3
amesh,all
agen,12,all,,,,,30
nummrg,node
nummrg,kp

/solu
antype,static
asel,s,loc,z,0
alls,below,area
nsla,s,1
sf,all,pres,1
```

```
asel,s,loc,x,1
alls,below,area
nsla,s,1
d,all,ux,0,,,,,uy,zu
pstres,on
alls
solve
fini

/solu
antype,buckle
bucopt,lanb,10
pstres,on
outres,all,all
alls
solve

/post1
set,first
*get,LOADF,active,,set,freq
fini
/prep7
alls
upgeom,0.02,1,1,luefter,rst
finish

/solu
antype,static
pstres,on
nlgeom,on
lnsrch,on
outres,all,all
nsubst,20,1e5,20
time,LOADF*1.1
sf scale,all,LOADF*1.1
solve
alls
save
finish

/post26
nsol,2,1648,u,y,uy_2
xvar,1
plvar,2
```

Berechnung von Strahlungsproblemen mit dem Radiositysolver

Ausgabe: 12 / 2002

Problem:



Berechnet werden soll die Strahlung zwischen zwei Körpern mit Hilfe der Radiosity Methode.

Erläuterung:

Im ANSYS gibt es neben der „alten“ Matrixmethode zur Berechnung der Strahlung die sogenannte Radiosity Methode. Dieses ist sehr viel einfacher zu handhaben und anwendbar auch für größere Strahlungsprobleme. Insbesondere dann, wenn bei einem der im Strahlungsaustausch stehen Körper das Temperaturniveau nicht durch entsprechende Randbedingungen definiert war, traten erhebliche Konvergenzprobleme auf. In Rev.7.0 ist das Lösungsverfahren erheblich verbessert worden, so dass auch solche Probleme jetzt effizient mit der Radiosity Methode gelöst werden können.

Beispiel:

Untersucht wird der Strahlungsaustausch zwischen einer Kugel und einer Platte, die jeweils mit Shell57 Elementen modelliert werden. Auf der Platte wird eine konstante Temperatur von 1000°C vorgegeben. Das Temperaturniveau der Kugel ist nicht durch weitere Randbedingungen spezifiziert, es wird indirekt über die Umgebungstemperatur (space temperature) des offenen Strahlungsraumes definiert. Es sind keine weiteren Lösungseinstellungen erforderlich, das Verfahren konvergiert in wenigen Iterationen.



Berechnung von Strahlungs- problemen mit dem Radiositysolver

Ausgabe: 12 / 2002

ANSYS Eingabesatz (ANSYS 7.0):

```
fini
/clear
/filna,kugel

/prep7
et,1,57
mp,kxx,1,1.25

r,1,0.1
sphere,0,1
wpof,,,-3
rect,-.5,.5,-.5,.5
esiz,.15

ames,all

/solu

!Emissivität
sfe,all,2,rdsf,1,0.8
!Enclosure
sfe,all,2,rdsf,2,1

!Stefan-Boltzmann-Konstante
stef,0.1713E-08
!Temperaturoffset
toffs,273

!Umgebungstemperatur 0°C
sptemp,1,0

!Temperatur auf die Fläche 1000°C
da,3,temp,1000

solve

/post1
set,last
plnsol,temp
```

Berechnung von Ausfallwahrscheinlichkeiten

Ausgabe: 12 / 2002

Problem:

Sämtliche Eingangsgrößen eines Berechnungsmodells streuen. Die in den üblicherweise verwendeten Normen enthaltenen Sicherheitsbeiwerte berücksichtigen eine angenommene Streuung der Widerstands und der Belastungsgrößen. Das Normenkonzept geht dann davon aus, dass ein bestimmter Prozentsatz der Belastungsgrößen die Widerstandsgrößen überschreiten darf. Es wird also eine geringe Ausfallwahrscheinlichkeit vorgegeben. Wenn man nun genaue Kenntnisse über die Streuung der Widerstandsseite hat und auch die Streuung der Belastungen aus statistischen Untersuchungen kennt, kann man mit dieser Information ein deutlich wirtschaftlicheres Bemessungskonzept für den Einzelfall aufstellen.

Erläuterung:

In ANSYS gibt es das Modul für Probabilistic Design. Dahinter verbirgt sich die Möglichkeit Parametern in einem Inputfile statistische Verteilungen zuzuweisen und deren Einfluss auf bestimmte Ausgangsgrößen zu berechnen. Dazu sind im PDS Modul verschiedene Algorithmen implementiert, die das Inputfile gemäss einer vorgegebenen Anzahl mehrmals mit entsprechend der Verteilungen bestimmten Eingangsgrößen aufrufen.

Dieses Vorgehen entspricht einer Versuchsreihe, die nicht im Labor, sondern virtuell durchgeführt wird. Dieses mehrmalige Ausführen einer Simulation wird als Monte Carlo Simulation bezeichnet und ist praktisch auf alle Berechnungen in ANSYS anwendbar.

Da diese Art der Berechnung sehr zeitaufwendig ist, gibt es in ANSYS die Möglichkeit (ohne weitere Lizenz) mehrere Rechner parallel zu nutzen. Darüberhinaus wird auch ein Response Surface Algorithmus angeboten, der auf Grundlage einiger weniger Monte Carlo Simulationen eine das Problem beschreibende Funktion berechnet, die dann innerhalb weniger Minuten in grossen Anzahlen gelöst werden kann. Die Response Surface ist aber nicht auf alle Probleme anwendbar. Dies muss im Einzelfall geprüft werden.

Für die Ausgangsgrößen kann man schliesslich anhand deren Verteilungen erkennen wieviele Bauteile in der Versuchsreihe ein bestimmtes Kriterium nicht erreicht haben.


Neben dieser Brechnungsmöglichkeit der Ausfallwahrscheinlichkeiten bietet das PDS Modul auch die Möglichkeit Sensitivitäten zu bestimmen, die den Benutzer in die Lage versetzen zu entscheiden auf welche Einagnsgrößen in der Produktion in Hinblick auf die Wirtschaftlichkeit besonders geachtet werden muss und welche weniger strengen Kontrollen zu unterziehen sind.

Beispiel:

Im vorliegenden Beispiel wird eine Platte mit Loch berechnet. Als Eingangsgrößen werden der E-Modul mit einer Gauss-Verteilung mit 5% Standardabweichung und der Lochradius mit einer gleichförmigen Verteilung von 1.8mm bis 2.3mm betrachtet. Die Ausgangsgrößen sind die maximale Verschiebung und die maximale Vergleichsspannung.

Als Ausfallkriterium wird eine Vergleichsspannung grösser der Fliessschnnung von 416N/mm² angenommen.

Dieses Kriterium wird von rund 6.5% der Versuchsergebnisse überschritten.



Berechnung von Ausfallwahrscheinlichkeiten

Ausgabe: 12 / 2002

ANSYS Eingabesatz (ANSYS 7.0) :

```
fini
/clear

!PDS-Beispiel Platte mit Loch
!Lösung mit Monte-Carlo Simulation
```

```
/eshape,1
```

```
*create,platte_ml,anl
em=2.1e5
rad=2.0
```

```
/prep7
mp,ex,1,em
mp,prxy,1,.3
```

```
et,1,181
r,1,.3
```

```
rectng,0,10,0,10
wpoff,5,5
pcirc,0,rad
aovlap,all
```

```
adele,2,,1
wprot,45,90
asbw,all
wprot,,90
asbw,all
esize,1
lcomb,13,17
lcomb,11,15
lcomb,7,12
lcomb,14,18
lsel,s,,5,9,4
lsel,a,,10,16,6
lesi,all,,6
mshkey,1
amesh,all
```

```
/solu
lsel,s,loc,x,0
lsel,a,loc,x,10
nsl,s,1
d,all,all,0
```

```
lsel,s,,7,11,4
lsel,a,,13,14
nsl,s,1
f,all,fz,-1
```

```
alls
solve
```

```
/post1
etab,seqvmax_,s,eqv
esort,etab,seqvmax_,0
*get,seqvmax,sort,,max
```

```
nsort,u,sum,0
*get,dmax,sort,,max
```

```
fini
*end
```

```
/inp,platte_ml,anl
/pds
pdanl,platte_ml,anl
```

```
pdvar,em,gaus,2.1e5,10500
pdvar,rad,unif,1.8,2.3
```

```
pdvar,seqvmax,response
pdvar,dmax,response
```

```
pdmeth,mcs,lhs
pdlhs,333,3
```

```
pdexe,platte_mit_loch
pdprob,platte_mit_loch,seqvmax,gt,416
```

Termine rund um CADFEM

Aktuelle Seminartermine:

- Update ANSYS 7.0 und Workbench Environment
09.-10.01.03 in Grafing bei München
- Schnuppertraining DesingSpace
20.01.03 in Grafing bei München
- Einführung in ANSYS Strukturmechanik
21.-24.01.03 in Grafing bei München
- FE-Idealisierung in der Strukturmechanik
21.-24.01.03 in in Leinfelden-Echterdingen bei Stuttgart
- Effizientes Arbeiten mit ANSYS
28.-31.01.03 in Grafing bei München
- Einführung in DesignSpace
28.-29.01.03 in Leinfelden-Echterdingen bei Stuttgart
- Update ANSYS 7.0 und Workbench Environment
30.-31.01.03 in Leinfelden-Echterdingen bei Stuttgart
- Einführung in die Temperaturfeldberechnung mit ANSYS
28.-31.01.03 in Burgdorf bei Hanover